

# Streifzüge durch die Grafikwelt (Teil 2)

**Grafik mit dem C 64, dem C 128 oder einem anderen Computer. Auf Bildschirm, Drucker oder Plotter. Unser Grafik-System nimmt Formen an.**

**D**ie in dieser und den kommenden Folgen gezeigten Programme sollen möglichst weitgehend unabhängig sein von der Art des verwendeten Grafik-Systems. Ob Sie also einen Commodore 64 mit HiRes-3 oder GBasic (oder...) benutzen und die Ausgabe auf dem Bildschirm, dem Drucker oder einem Plotter stattfinden lassen, ob Sie zu den Glücklichen gehören, die schon einen Commodore C 128 vor sich stehen haben... all das soll möglichst weitgehend erfaßt sein. Nicht immer — gestehe ich ganz freimütig — läßt sich so eine Allgemeingültigkeit erreichen. Besonders in dieser Folge werden die drei größeren Programme noch spezialisiert sein. Die kleineren aber — die fürs Auge — sollen diese Anforderungen schon erfüllen.

Dazu werde ich allgemein verständliche Befehlsworte verwenden (wie zum Beispiel LINIE oder PUNKT etc.) und deren Entsprechung in zwei Grafik-Systemen angeben: In HiRes-3-Syntax und in der Syntax, die mit dem Plotter 1520 verwendet wird. Falls Sie ein anderes System benutzen oder einen C 128 Ihr eigen nennen, dürfte es Ihnen nicht schwerfallen, die Syntax-Anpassungen vorzunehmen. Folgende zehn Befehle sollen zu nächst vorgestellt werden (wobei sich später noch der eine oder andere zusätzliche Befehl als nötig herausstellen kann):

**INIT**

Werde ich immer dann verwenden, wenn die Grafik initiali-

siert werden soll. Damit wird die Grafik eingeschaltet, eine eventuell Bit-Map eingerichtet und gelöscht und die Farbgebung bestimmt.

**START**

Soll das Grafiksystem in einen definierten Ausgangszustand bringen.

**MITTE**

Legt den Koordinatenursprung in die Bildmitte.

**PUNKT(X,Y)**

Setzt einen Punkt an die Stelle X,Y.

**LINIE(XA,YA,XB,YB)**

Zieht eine Linie vom Ausgangspunkt XA,YA zum Endpunkt XB,YB.

**KREIS(XM,YM,RX,RY)**

Zeichnet eine Ellipse (RX ungleich RY) oder einen Kreis (RX = RY) um den Mittelpunkt XM, YM mit den Radien RX und RY.

**TEXT(A\$,XT,YT)**

Schreibt den Text A\$ ab der Bildstelle XT,YT.

**SHOW**

Ein erstelltes Bild wird gezeigt.

**NORMAL**

Schaltet den Grafik-Modus ab.

**GRESET**

Führt einen Grafik-System-Reset aus. Danach muß für erneute Benutzung der Grafik wieder neu initialisiert werden.

In der Tabelle 1 sind diese allgemeinen Befehle aufgeführt zusammen mit den Übersetzungen in die Syntax von HiRes-3 und die des Plotters 1520:

Sehen wir uns das nun an zwei netten Beispielen an. Als erstes das Programm »verschlungene Spiralen«, welches (siehe Bilder 1a, 1b) mehrere Spiralen (nämlich N), die mit einem Startwinkel W aus dem Zentrum heraus treten, zu einer Scheibe mit dem Radius R anordnet.



**Bild 1a. Verschlungene Spiralen mit HiRes-3**



**Bild 1b. Dasselbe mit dem Plotter 1520**

| Wort               | HIRES-3-Syntax                                                                                | Plotter-Syntax                                                                                                                                                                                                                                                                     |
|--------------------|-----------------------------------------------------------------------------------------------|------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| INIT               | POKE53280,0<br>(Rahmen schwarz)<br>SYS37498 (HIRES anschalten)<br>HFL,6,12<br>(blau auf grau) | OPEN1,6,1 (Plot X,Y)                                                                                                                                                                                                                                                               |
| START              | TRS,0,320,0,200<br>legt Ursprung nach links unten<br>x von 0 bis 320<br>y von 0 bis 200       | OPEN2,6,2 (Stiftfarbe)<br><br>PRINT # 2,0 (schwarz)<br>CLOSE2<br>PRINT # 1,"M",0,200<br><br>PRINT # 1,"I"<br>Papiervorschub um 200 Einheiten. Dort auch Koordinatenursprung.                                                                                                       |
| MITTE              | TRS,-160,160,-100,100<br><br>Ursprung in Bildschirmmitte                                      | PRINT # 1,"R",240,0<br>PRINT # 1,"I"<br><br>Ursprung in Papiermitte.                                                                                                                                                                                                               |
| PUNKT(X,Y)         | TPK,X,Y                                                                                       | PRINT # 1,"R",X,Y<br>PRINT # 1,"I",X+2,Y+2<br>PRINT # 1,"R",X,Y<br>etwas ungenau, weil einzelner Punkt nicht sichtbar. Deshalb sehr kleiner Strich.                                                                                                                                |
| LINIE(XA,YA,XB,YB) | TLN,XA,YA,XB,YB                                                                               | PRINT # 1,"R",XA,YA<br>PRINT # 1,"I",XB,YB<br>Hier braucht man ein kleines Programm:<br>M=50:D=360*(PI/(M*180))<br>DIMT(M):T(0)=2*PI<br>FOR I=1 TO M:T(I)=T(I-1)+D<br>PRINT # 1,"R",RX*COS(T(I-1))+XM,RY*SIN(T(I-1))+YM<br>PRINT # 1,"I",RX*COS(T(I))+XM,RY*SIN(T(I))+YM<br>NEXT I |
| KREIS(XM,YM,RX,RY) | TKR,XM,YM,RX,RY,2*PI                                                                          |                                                                                                                                                                                                                                                                                    |
| TEXT(A\$,XT,YT)    | TEX,A\$,YT,XT                                                                                 | OPEN4,6:PRINT # 1,"R",XT,YT<br>PRINT # 4,A\$:CLOSE4<br>PRINT # 1,"M",240,YT<br>PRINT # 1,"I"                                                                                                                                                                                       |
| SHOW               | HAN                                                                                           | PRINT # 1,"R",0,200<br>Papiervorschub um 200 Einheiten                                                                                                                                                                                                                             |
| NORMAL             | HOF                                                                                           | CLOSE1                                                                                                                                                                                                                                                                             |
| GRESET             | LOE:AUS                                                                                       | OPEN7,6,7:PRINT # 7:<br>CLOSE7                                                                                                                                                                                                                                                     |

**Tabelle 1. Die grundlegenden allgemeinen Grafik-Befehle und ihre Übersetzung in HiRes-3- oder Plotter-Syntax**

Das Programm »Spiralen« (Listing 1) verwendet die allgemeinen Grafik-Befehle. Dieses Listing kann jedoch nicht auf dem C 64 laufen. Aber je nachdem, welchen Computer Sie verwenden, oder welche Basic-Erweiterung, können Sie deren Befehle hier einsetzen. Listing 2 und Listing 3 geben zwei Beispiele.

Probieren Sie nun mal aus — unter Zuhilfenahme der Tabelle 1 — anstelle der allgemeinen Befehle die HiRes-3-Syntax einzusetzen. Dabei sollte dann etwas Ähnliches herauskommen wie Listing 2, HiRes-Spiralen:

Unter Verwendung der Plotter-Syntax entsteht das Programm Plotter-Spiralen (Listing 3).

Sollten Sie ein anderes Grafik-System benutzen, können Sie solch ein Programm auf ähnlich simple Weise übersetzen.

Als zweites Beispiel soll das Programm »Fensterrose« dienen

(der Name rührt daher, daß das Ergebnis aussieht, wie die Fensterrosen in alten gotischen Kirchen). Auch hier biete ich Ihnen drei Versionen: »Fensterrose« ist das mit den allgemeinen Grafik-Befehlen (Listing 4), »HiRes-Fensterrose« ist die HiRes-3-Fassung (Listing 5) und »Plotter-Fensterrose« schließlich die Plotter-Version (Listing 6).

Die Ergebnisse finden Sie als Bilder 2a und 2b:

Die Fensterrose-Programme fragen nach der Anzahl der Stützpunkte auf dem Kreisumfang, dann nach dem gewünschten Radius und schließlich noch nach dem Ort, an den der Text gedruckt werden soll. Wie Sie feststellen werden, ist Fensterrose sehr vielseitig, und auf dem Bildschirm lassen sich in der HiRes-Version Moiree-Muster erzeugen.

**1. Eine »Business-Anwendung«**

Eines der Themen, auf die wir

in den nächsten Folgen zu sprechen kommen werden, ist die sogenannte Präsentationsgrafik, die häufig im kaufmännischen Bereich Verwendung findet. Darunter versteht man zum Beispiel Balkendiagramme oder die vielgeliebten Tortendiagramme. Das hier gezeigte Beispiel arbeitet mit HiRes-3 und unter Umständen einem Drucker. Es bedient sich eines sogenannten Abweichungs-Balkendiagrammes. Von maximal 50 eingegebenen Werten (Umsatzziffern, Gewinne, etc.) stellt es den Mittelwert als horizontale Linie dar und daran dann die positiven oder negativen Abweichungen der einzelnen Eingabebeträge. Außerdem gibt das Programm Auskunft über den Zahlenwert des Mittels und die maximalen Abweichungen. Ein Beispiel für solch ein Diagramm sehen Sie in Bild 3:

Das Programm »Abweichungen« (Listing 7) erfragt zunächst die Anzahl der Werte, dann jeden Einzelwert. Es ist leicht zu ändern, so daß es die Werte auch von der Diskette oder Kassette nehmen kann. Am Ende der grafischen Darstellung können Sie durch »D« einen Ausdruck auf dem Drucker 1526 (oder kompatible) erzielen, durch »N« eine neue Eingabe starten oder durch »E« das Programm beenden.

## 2. Ein Plotter-Programm

Grafik und Mathematik sind nicht voneinander zu trennen. Eine häufige Anwendung ist die Untersuchung von 2D-Funktionen durch die grafische Darstellung. Die Zeiten der mühselig berechneten Wertetabellen und der Kurvendiskussion sind zwar

noch nicht ganz vorbei (es gibt halt doch noch einige exotische Funktionen, die dem Computer widerstehen), für die weitaus meisten Anwendungen ist aber das hier vorgestellte Programm »Funktionenplot« (Listing 8) leistungsfähig genug. Es bedarf dazu eines Plotters 1520. Zunächst wird vorgestellt, welche Funktion sich derzeit im Programm befindet. Man hat nun die Möglichkeit, eine neue Funktion einzuspeisen. Danach bestimmt man selbst, wie das Koordinatensystem aussehen soll (bei einer Ausdehnung von mehr als 25 Einheiten in X-Richtung wird keine Skalierung mehr vorgenommen) und man bestimmt, in welchem Teil des Systems die Funktion darzustellen ist. Ein Beispiel für die entstehenden Plots zeigt Bild 4.

Es gibt noch eine Menge Raffinessen, die in diesem Beispielprogramm fehlen. Aber die Untersuchung von Funktionen wird uns ebenfalls eine der kommenden Grafik-Folgen beschäftigen. Dort sollen dann auch alle Feinheiten zur Sprache kommen.

Fortsetzung auf Seite 170




Bild 2a. Der Matrixdrucker gibt es nur undeutlich wieder: Fensterrose mittels HiRes-3






Bild 2b. Von bestechender Schönheit: Die Fensterrose mit dem Plotter gezeichnet

### ABWEICHUNGS-BALKENDIAGRAMM



GROESSTE UEBERSCHREITUNG = 16.1666667  
GROESSTE UNTERSCHREITUNG = -35.8333333  
MITTELWERT = 1.8333333

Bild 3. Das Abweichungs-Balken-Diagramm



$$Y = \sin(X)$$

Bild 4. Ein Beispiel zur Arbeit des Plotter-Programmes zur Darstellung von Funktionen




Bild 5. HiRes-3 zeichnet 3D-Funktionen

```
1 REM*****
2 REM*
3 REM* VERSCHLUNGENE SPIRALEN *
4 REM* VERSION M. ALLGEMEINEN BEFEHLEN*
5 REM*
6 REM* HEIMO PONNATH HAMBURG 1985 *
7 REM*
8 REM*****
9 REM ***** EINGABEN *****
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
   INT CHR$(147)
15 INPUT"ANZAHL SPIRALEN";N:INPUT"TEXTSTAR
   T XT,YT";XT,YT
20 INPUT"RADIUS,STARTWINKEL";R,A1
22 INIT
25 START
30 MITTE
```

Listing 1. Spiralen  
Das Spiralen-Programm mit den allgemeinen Grafik-Befehlen

```

35 GOSUB 100:KREIS(0,0,R,R)
40 A$="VERSCHLUNGENE SPIRALEN"
45 TEXT(A$,XT,YT)
50 SHOW
55 POKE 198,0:WAIT 198,1
60 N OR MAL
65 GRESET
70 END
100 REM***** UNTERPROGRAMM ZEICHNEN ***
102 NN=N*100:RA=R/NN:DIM T(NN):A=A1
105 FOR J=1 TO N:A=A+2*PI/N
110 T=A:TH=2*PI/100:T(0)=T
115 FOR I=1 TO NN:T(I)=T(I-1)+TH:RR=RA*I
120 LINIE((RR-RA)*COS(T(I-1)),(RR-RA)*SIN(
  T(I-1)),RR*COS(T(I)),RR*SIN(T(I)))
125 NEXT I
130 NEXT J
135 RETURN

```

© 64'er

## Listing 1. Spiralen (Schluß)

```

1 REM***** <038>
2 REM* * <051>
3 REM* VERSCHLUNGENE SPIRALEN * <232>
4 REM* VERSION FUER H I R E S - 3 * <066>
5 REM* * <054>
6 REM* HEIMO PONNATH HAMBURG 1985 * <131>
7 REM* * <056>
8 REM***** <045>
9 REM ***** EINGABEN ***** <180>
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
  INT CHR$(147) <124>
15 INPUT"ANZAHL SPIRALEN";N:INPUT"TEXTSTAR
  T XT,YT";XT,YT <253>
20 INPUT"RADIUS,STARTWINKEL";R,A1 <099>
22 REM***** INIT ***** <173>
23 SYS 37498:HFL,6,12 <047>
25 REM***** START ***** <009>
26 TRS,0,320,0,200 <237>
30 REM***** MITTE ***** <182>
32 TRS,-160,160,-100,100 <002>
35 GOSUB 100:TKR,0,0,R,R,2*PI <153>
40 A$="VERSCHLUNGENE SPIRALEN" <241>
42 REM***** TEXT ***** <092>
45 TEX,A$,YT,XT <179>
50 REM***** SHOW (HIER UNNOETIG)***** <021>
55 POKE 198,0:WAIT 198,1 <027>
60 REM***** NORMAL ***** <103>
62 HOF <061>
65 REM***** GRESET ***** <138>
67 LOE:AUS <214>
70 END <072>
100 REM***** UNTERPROGRAMM ZEICHNEN *** <048>
102 NN=N*100:RA=R/NN:DIM T(NN):A=A1 <033>
105 FOR J=1 TO N:A=A+2*PI/N <085>
110 T=A:TH=2*PI/100:T(0)=T <211>
115 FOR I=1 TO NN:T(I)=T(I-1)+TH:RR=RA*I <048>
120 TLN,(RR-RA)*COS(T(I-1)),(RR-RA)*SIN(T(
  I-1)),RR*COS(T(I)),RR*SIN(T(I)) <198>
125 NEXT I <209>
130 NEXT J <222>
135 RETURN <193>

```

© 64'er

Listing 2. HiRes-Spiralen.  
Die Übersetzung in die Syntax von HiRes-3

```

1 REM***** <038>
2 REM* * <051>
3 REM* VERSCHLUNGENE SPIRALEN * <232>
4 REM* VERSION FUER PLOTTER 1520 * <131>
5 REM* * <054>
6 REM* HEIMO PONNATH HAMBURG 1985 * <131>
7 REM* * <056>
8 REM***** <045>
9 REM ***** EINGABEN ***** <180>
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
  INT CHR$(147) <124>
15 INPUT"ANZAHL SPIRALEN";N:INPUT"TEXTSTAR

```

Listing 3. Plotter-Spiralen  
Das Spiralen-Programm mit der Plotter-Syntax

```

T XT,YT";XT,YT <253>
20 INPUT"RADIUS,STARTWINKEL";R,A1 <099>
22 REM***** INIT ***** <173>
23 OPEN 1,6,1:OPEN 2,6,2:PRINT#2,0:CLOSE 2 <236>
25 REM***** START ***** <009>
26 PRINT#1,"M",0,-200:PRINT#1,"I" <237>
30 REM***** MITTE ***** <182>
32 PRINT#1,"R",240,0:PRINT#1,"I":GOSUB 100 <177>
33 REM ***** KREIS ***** <101>
35 M=50:D=360*PI/(M*180):DIM K(M):K(0)=2*PI <212>
36 FOR I=1 TO M:K(I)=K(I-1)+D:PRINT#1,"R",
  R*COS(K(I-1)),R*SIN(K(I-1)) <194>
37 PRINT#1,"J",R*COS(K(I)),R*SIN(K(I)):NEX
  T I:PRINT#1,"H" <130>
40 A$="VERSCHLUNGENE SPIRALEN" <241>
42 REM***** TEXT ***** <092>
43 OPEN 4,6:PRINT#1,"R",XT,YT:PRINT#4,A$:C
  LOSE 4 <171>
45 PRINT#1,"M",240,-YT:PRINT#1,"I" <125>
50 REM***** SHOW ***** <224>
52 PRINT#1,"R",0,-200:PRINT#1,"I" <072>
55 POKE 198,0:WAIT 198,1 <027>
60 REM***** NORMAL ***** <103>
62 PRINT#1,"H":CLOSE 1 <145>
65 REM***** WERTE AUSDRUCKEN ***** <073>
70 OPEN 3,6,3:PRINT#3,0:OPEN 4,6:PRINT#4:P
  RINT#4," N","{2SPACE}XT","{2SPACE}YT","
  {2SPACE}R","{2SPACE}W" <251>
75 PRINT#4,N,XT,YT,R,A1:PRINT#3,1:PRINT#4:
  CLOSE 4:CLOSE 3 <042>
80 REM***** GRESET ***** <153>
85 OPEN 7,6,7:PRINT#7:CLOSE 7 <194>
90 END <092>
100 REM***** UNTERPROGRAMM ZEICHNEN *** <048>
102 NN=N*100:RA=R/NN:DIM T(NN):A=A1 <033>
105 FOR J=1 TO N:A=A+2*PI/N <085>
110 T=A:TH=2*PI/100:T(0)=T <211>
115 FOR I=1 TO NN:T(I)=T(I-1)+TH:RR=RA*I <048>
120 PRINT#1,"R", (RR-RA)*COS(T(I-1)), (RR-RA
  )*SIN(T(I-1)) <201>
122 PRINT#1,"J",RR*COS(T(I)),RR*SIN(T(I)) <047>
125 NEXT I <209>
130 NEXT J <222>
135 RETURN <193>

```

© 64'er

## Listing 3. Plotter-Spiralen (Schluß)

```

1 REM ***** <038>
2 REM * * <051>
3 REM * FENSTERROSE * <232>
4 REM * VERSION M. ALLGEM. BEFEHLEN * <131>
5 REM * * <054>
6 REM * HEIMO PONNATH HAMBURG 1985 * <131>
7 REM * * <056>
8 REM ***** <045>
9 REM ***** EINGABEN ***** <180>
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
  INT CHR$(147) <124>
15 INPUT"ANZAHL STUETZPUNKTE";N
20 INPUT"RADIUS";R:INPUT"TEXTORT XT,YT";XT
  ,YT
25 INIT
35 START
45 MITTE
60 D=2*PI/N:DIM X(N),Y(N)
65 KREIS(0,0,R,R)
80 GOSUB 200
90 A$="FENSTERROSE"
100 TEXT(A$,XT,YT)
105 SHOW
120 POKE 198,0:WAIT 198,1
125 N OR MAL
135 GRESET
160 END
200 REM ***** UP ZEICHNEN *****
210 FOR I=1 TO N:T=T+D:X(I)=R*COS(T):Y(I)=
  R*SIN(T):NEXT I
220 S=N-1
230 FOR I=1 TO S:Z=I+1
240 FOR J=Z TO N
245 LINIE(X(I),Y(I),X(J),Y(J))
250 NEXT J:NEXT I:RETURN

```

© 64'er

Listing 4. Fensterrose. Ein gotisches Rundfenster,  
programmiert mit allgemeinen Grafik-Befehlen

```

1 REM ***** <051>
2 REM * * <051>
3 REM * FENSTERROSE * <087>
4 REM * VERSION FUER H I R E S 3 * <248>
5 REM * * <054>
6 REM * HEIMO PONNATH HAMBURG 1985 * <131>
7 REM * * <056>
8 REM ***** <058>
9 REM **** EINGABEN ***** <162>
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
INT CHR$(147) <124>
15 INPUT"ANZAHL STUETZPUNKTE";N <068>
20 INPUT"RADIUS";R:INPUT"TEXTORT XT,YT";XT
,YT <130>
25 REM **** INIT ***** <197>
30 SYS 37498:HFL,6,12 <054>
35 REM **** START ***** <061>
40 TRS,0,320,0,200 <251>
45 REM **** MITTE ***** <239>
50 TRS,-160,160,-100,100 <064>
55 REM***** <202>
60 D=2*PI/N:DIM X(N),Y(N) <227>
65 REM **** KREIS ***** <049>
70 TKR,0,0,R,2*PI <122>
75 REM***** <222>
80 GOSUB 200 <032>
90 A$="FENSTERROSE" <005>
95 REM **** TEXT ***** <166>
100 TEX,A$,YT,XT <234>
105 REM **** SHOW (HIER UNDOETIG) ***** <145>
115 REM***** <178>
120 POKE 198,0:WAIT 198,1 <092>
125 REM **** NORMAL ***** <252>
130 HOF <129>
135 REM **** GRESET ***** <036>
150 LOE:AUS <041>
160 END <162>
200 REM ***** UP ZEICHNEN ***** <054>
210 FOR I=1 TO N:T=T+D:X(I)=R*COS(T):Y(I)=
R*SIN(T):NEXT I <127>
220 S=N-1 <099>
230 FOR I=1 TO S:Z=I+1 <043>
240 FOR J=Z TO N <014>
242 REM **** LINIE ***** <252>
245 TLN,X(I),Y(I),X(J),Y(J) <186>
247 REM***** <050>
250 NEXT J:NEXT I:RETURN <070>

```

© 64'er Listing 5. Hires-Fensterrose.  
Fensterrose, übersetzt in die HiRes-3-Syntax

```

1 REM ***** <051>
2 REM * * <051>
3 REM * FENSTERROSE * <087>
4 REM * VERSION FUER PLOTTER 1520 * <131>
5 REM * * <054>
6 REM * HEIMO PONNATH HAMBURG 1985 * <131>
7 REM * * <056>
8 REM ***** <058>
9 REM **** EINGABEN ***** <162>
10 POKE 53280,0:POKE 53281,0:POKE 646,5:PR
INT CHR$(147) <124>
15 INPUT"ANZAHL STUETZPUNKTE";N <068>
20 INPUT"RADIUS";R:INPUT"TEXTORT XT,YT";XT
,YT <130>
25 REM **** INIT ***** <197>
30 OPEN 1,6,1:OPEN 2,6,2:PRINT#2,1:CLOSE 2 <251>
35 REM **** START ***** <061>
40 PRINT#1,"M",0,-200:PRINT#1,"I" <251>
45 REM **** MITTE ***** <239>
50 PRINT#1,"R",240,0:PRINT#1,"I" <200>
55 REM***** <202>
60 D=2*PI/N:DIM X(N),Y(N) <227>
65 REM **** KREIS ***** <049>
70 M=50:DD=360*PI/(M*180):DIM T(M):T(0)=2*PI <121>
72 FOR I=1 TO M:T(I)=T(I-1)+DD:PRINT#1,"R"
,R*COS(T(I-1)),R*SIN(T(I-1)) <236>
74 PRINT#1,"J",R*COS(T(I)),R*SIN(T(I)):NEXT
I <173>
75 REM***** <222>
80 GOSUB 200 <032>
90 A$="FENSTERROSE" <005>
95 REM **** TEXT ***** <166>

```

Listing 6. Plotter-Fensterrose.  
Die Übersetzung in die Plotter-Sprache

```

100 OPEN 4,6:PRINT#1,"R",XT,YT:PRINT#4,A$:
CLOSE 4:PRINT#1,"M",240,-YT:PRINT#1,"I"
" <067>
105 REM **** SHOW ***** <044>
110 PRINT#1,"R",0,-200 <210>
115 REM***** <178>
120 POKE 198,0:WAIT 198,1 <092>
125 REM **** NORMAL ***** <252>
130 OPEN 3,6,3:PRINT#3,0:OPEN 4,6:PRINT#4:
PRINT#4," N","{3SPACE}R","{4SPACE}XT",
"{4SPACE}YT" <001>
132 PRINT#4,N,R,XT,YT:PRINT#3,1:PRINT#4:CL
OSE 3:CLOSE 4:CLOSE 1 <167>
135 REM **** GRESET ***** <036>
150 OPEN 7,6,7:PRINT#7:CLOSE 7 <003>
160 END <162>
200 REM ***** UP ZEICHNEN ***** <054>
210 FOR I=1 TO N:T=T+D:X(I)=R*COS(T):Y(I)=
R*SIN(T):NEXT I <127>
220 S=N-1 <099>
230 FOR I=1 TO S:Z=I+1 <043>
240 FOR J=Z TO N <014>
242 REM **** LINIE ***** <252>
245 PRINT#1,"R",X(I),Y(I):PRINT#1,"J",X(J)
,Y(J) <090>
247 REM***** <050>
250 NEXT J:NEXT I:RETURN <070>

```

© 64'er Listing 6. Plotter-Fensterrose (Schluß)

```

1 REM ***** <132>
2 REM * * <051>
3 REM * ABWEICHUNGSBALKENDIAGRAMM * <193>
4 REM * MITTELS HIRES-3 * <078>
5 REM * * <054>
6 REM * HEIMO PONNATH HAMBURG 1985 * <131>
7 REM * * <056>
8 REM * HIRES-3 MUSS IM SPEICHER SEIN * <034>
9 REM ***** <140>
10 REM <072>
15 REM ++++++ INITIALISIEREN ++++++ <173>
20 REM <082>
25 POKE 52,112:POKE 56,112:SYS 37498:GOTO
65 <015>
30 REM <092>
35 REM ++++++ UP CURSOR SETZEN ++++++ <005>
40 REM <102>
45 POKE 211,SP:POKE 214,ZL:SYS 58640:RETUR
N <053>
50 REM <112>
55 REM ++++++ EINGABETEIL ++++++ <189>
60 REM <122>
65 PRINT CHR$(147):SYS 34647:POKE 646,14:Z
L=10:SP=5:GOSUB 45 <207>
70 PRINT"BALKENDIAGRAMM WELCHES DEN":ZL=12
:GOSUB 45 <090>
75 PRINT"MITTELWERT UND DIE ABWEICHUNGEN":
ZL=14:GOSUB 45 <254>
80 PRINT"GRAFISCH DARSTELLT.":PAU,5:PRINT
CHR$(147):ZL=10:SP=7:GOSUB 45 <011>
85 PRINT"WIEVIELE WERTE (MAX=50) ";:INPUT
N <159>
90 DIM W(N),A(N):PRINT CHR$(147) <133>
95 MA=0:MI=0 <113>
100 FOR I=1 TO N:PRINT"WERT NUMMER "I="";:
INPUT W(I):S=S+W(I):NEXT I <067>
105 REM <167>
110 REM + BERECHNUNG DER ABWEICHUNGEN + <197>
120 REM <182>
125 M=S/N:FOR I=1 TO N:A(I)=W(I)-M:IF A(I)
>MA THEN MA=A(I) <150>
130 IF A(I)<MI THEN MI=A(I) <088>
135 NEXT I <219>
140 REM <202>
145 REM ++++++ BILDSCHIRMAUFTEILUNG ++++++ <006>
150 REM <212>
155 Q=MA+ABS(MI):Q1=.06*Q:Q2=.2*Q:M1=MA+Q1
:M2=MI-Q2 <201>
160 DX=INT(320/N):TRS,-5,319,M2,M1 <195>
165 REM <227>
170 REM ++++++ ZEICHNEN ++++++ <241>
175 REM <237>

```

Listing 7. Abweichungen. Ermittelt und zeichnet den Durchschnitt  
und die Abweichungen davon: Das Programm Abweichungen

```

180 HFL,14,6:TLN,-1,0,319,0:SYS 35256:X=0 <153>
185 FOR I=1 TO N:IF A(I)>0 THEN:TRX,X,A(I)
,X+DX,0:GOTO 195 <000>
190 TBK,X,0,X+DX,A(I) <028>
195 X=X+DX:NEXT I <052>
200 ZL=0:SP=5:GOSUB 45:PRINT"ABWEICHUNGS-B
ALKENDIAGRAMM" <238>
205 ZL=21:SP=1:GOSUB 45:PRINT"MITTELWERT =
"M:ZL=22:GOSUB 45 <177>
210 PRINT"+ MAX "MA,"- MAX "MI <023>
215 ZL=23:GOSUB 45:PRINT"E = ENDE","N = NE
U","D = DRUCKEN" <172>
220 REM <026>
225 REM +++++ MENUE-AUSWERTUNG +++++ <100>
230 REM <036>
235 GET A$:IF A$<>"E"AND A$<>"N"AND A$<>"D
"THEN 235 <169>
240 REM <046>
245 REM ----- OPTION NEUE WERTE ----- <135>
250 REM <056>
255 IF A$="N"THEN SYS 35377:HOF:RUN <080>
260 REM <068>
265 REM ----- OPTION DRUCKEN ----- <141>
270 REM <078>
275 IF A$="D"THEN GOSUB 310:HOF:PRINT CHR$(
147):GOTO 215 <099>
280 REM <088>
285 REM ----- OPTION PROGRAMMENDE ----- <116>
290 REM <098>
295 SYS 35377:HOF:END <073>
300 REM <108>
305 REM ----- OPTION DRUCKEN UP ----- <147>
310 REM <118>
315 SYS 35377:OPEN 1,4,10:PRINT#1:CLOSE 1:
OPEN 1,4 <033>
320 PRINT#1,"ABWEICHUNGS-BALKENDIAGRAMM":S
YS 34865 <002>
325 PRINT#1,"GROESSTE UEBERSCHREITUNG = "M
A <058>
330 PRINT#1,"GROESSTE UNTERSCHREITUNG = "M
I <222>
335 PRINT#1,"MITTELWERT = "M:PRINT#1:CLOSE
1:RETURN <122>
340 REM <148>
345 REM ***** DAS WARS ! ***** <219>

```

© 64'er

## Listing 7. Abweichungen (Schluß)

```

1 REM ***** <132>
2 REM * <051>
3 REM * PLOTTER BELIEBIGER FUNKTIONEN * <097>
4 REM * MIT DEM PRINTER/PLOTTER 1520 * <118>
5 REM * <054>
6 REM * HEIMO PONNATH HAMBURG 1985 * <131>
7 REM * <056>
8 REM ***** <139>
9 GOTO 25 <019>
10 REM ++++++ UP CURSOR SETZEN ++++++ <077>
15 POKE 211,SP:POKE 214,Z:SYS 58640:RETURN <027>
20 REM ++++++ HAUPTPROGRAMM TEIL 1 ++ <228>
25 PRINT CHR$(147):Z=10:SP=1:GOSUB 15 <141>
30 PRINT"IM PROGRAMM BEFINDET SICH DIE FUN
KTION:" <130>
35 K=1:GOSUB 90:K=0:PRINT:PRINT"Y= "F$:Z=1
8:GOSUB 15 <042>
40 PRINT CHR$(18)"A"CHR$(146)"LTE ODER "CH
R$(18)"N"CHR$(146)"EUE FUNKTION ?" <212>
45 GET A$:IF A$<>"A"AND A$<>"N"THEN 45 <207>
50 IF A$="A"THEN 85 <036>
55 REM ++ NEUE FUNKTION INS PROGRAMM ++ <021>
60 Z=20:SP=3:GOSUB 15:PRINT"NEUE FUNKTION:
":INPUT"Y= "F$:POKE 646,6 <028>
65 PRINT CHR$(147)CHR$(17)CHR$(17)"90F$="C
HR$(34)F$CHR$(34) <158>
70 PRINT"95DEFN A(X)="F$:PRINT"RUN85":PRIN
T CHR$(19); <251>
75 POKE 631,13:POKE 632,13:POKE 633,13:POK
E 198,3:END <033>
80 REM ++++++ HAUPTPROGRAMM TEIL 2 ++ <048>
85 PRINT CHR$(147):POKE 646,14 <213>
90 F$="X^2/2" <177>
95 DEF FN A(X)=X^2/2 <118>
100 IF K=1 THEN RETURN <109>
105 REM +++++ VARIABLE UND FUNKTIONEN ++ <240>
110 DEF FN T(X)=INT(479/(X0-XU)*X) <192>
115 X=0:Z=0:SP=0:XU=-1:X0=5:YU=-1:YO=5:YM=

```

```

6:X=XU:X=XO <143>
120 X1=0:X2=0:Y1=0:Y2=0:D=150:DY=.1:I=10:M
=10:DX=.2 <228>
125 A$="" <154>
130 REM +++ PLOTTERKANAELE OEFFNEN +++ <162>
135 OPEN 1,6,1:OPEN 4,6:OPEN 2,6,2:OPEN 3,
6,3:PRINT#2,3 <252>
140 REM +++++ PARAMETEREINGABEN +++++ <081>
145 Z=0:SP=0:GOSUB 15:PRINT"FUNKTION Y="F$ <231>
150 Z=2:GOSUB 15:PRINT"(FALLS SIE MIT UNST
ETIGKEITEN RECHNEN," <164>
155 PRINT"DANN VERMEIDEN SIE ES, DIE POLST
ELLEN" <041>
160 PRINT"ALS PARAMETER (XU,XO,EVTL.XA ODE
R XE) " <145>
165 PRINT"ZU VERWENDEN. D SOLLTE DANN KLEI
N SEIN." <248>
170 Z=7:SP=10:GOSUB 15:PRINT"XU UND XO =" <255>
175 Z=14:SP=10:GOSUB 15:PRINT"YU UND YO =" <032>
180 Z=16:SP=0:GOSUB 15:PRINT"SPEZ. ZEICHEN
BEREICH GEWUENSCHT (J/N)?" <228>
182 Z=22:SP=10:GOSUB 15:PRINT"SCHRITTWEITE
(D=CA.150)" <065>
185 Z=7:SP=25:GOSUB 15:INPUT XU,XO <233>
190 YM=INT(998*(XO-XU)/479):Z=9:SP=3:GOSUB
15:PRINT"YU MINIMUM = ",YM/2 <005>
195 Z=10:SP=3:GOSUB 15:PRINT"YO MAXIMUM =
",YM/2:Z=11:SP=0:GOSUB 15 <242>
200 PRINT"(DIE SUMME AUS YU,YO UND YO DARF
"YM:PRINT" NICHT UEBERSCHREITEN!) <239>
205 Z=14:SP=25:GOSUB 15:INPUT YU,YO:IF (YO-
YU)>YM THEN 205 <255>
210 Z=16:SP=36:GOSUB 15:INPUT A$:IF A$<>"J
"AND A$<>"N"THEN 210 <016>
215 IF A$="N"THEN XA=XU:XE=XO:GOTO 232 <149>
220 Z=18:SP=3:GOSUB 15:PRINT"(XA > XU UND
XE < XO !" <128>
225 Z=20:SP=10:GOSUB 15:PRINT"XA UND XE ="
:SP=25:GOSUB 15:INPUT XA,XE <151>
230 IF XA<XU OR XE>XO THEN 225 <046>
232 Z=22:SP=32:GOSUB 15:INPUT D <040>
235 REM +++ PLOTTER ANFANGSWERTE +++++ <247>
240 PRINT#1,"H":PRINT#1,"M",FN T(-XU),-FN
T(YO):PRINT#1,"I" <214>
245 X1=FN T(XU)+1:X2=FN T(XO)-1:Y1=FN T(YO
)-1:Y2=FN T(YU)+1 <129>
250 REM +++ PLOTTER KOORDINATENSYSTEM + <093>
255 PRINT#1,"R",0,Y1:PRINT#1,"J",0,Y2:PRIN
T#1,"R",X2,0:PRINT#1,"J",X1,0 <046>
260 REM +++ PLOTTER SKALIERUNG +++++ <144>
265 DY=(XO-XU)/47 <249>
270 FOR I=XU+1 TO XO-1:M=INT(I):PRINT#1,"R
",FN T(M),FN T(0) <022>
275 PRINT#1,"J",FN T(M),FN T(-DY) <182>
280 NEXT I <110>
285 FOR I=YU+1 TO YO-1:M=INT(I):PRINT#1,"R
",FN T(0),FN T(M) <115>
290 PRINT#1,"J",FN T(DY),FN T(M) <121>
295 NEXT I:PRINT#2,0 <005>
300 REM *** PLOTTER FUNKTION +++++ <025>
305 IF FN T(FN A(XA))>Y1 THEN PRINT#1,"R",
FN T(XA),Y1:GOTO 320 <240>
310 IF FN T(FN A(XA))<Y2 THEN PRINT#1,"R",
FN T(XA),Y2:GOTO 320 <183>
315 PRINT#1,"R",FN T(XA),FN T(FN A(XA)) <077>
320 FOR X=XA TO XE STEP (XO-XU)/D <174>
325 IF FN T(FN A(X))>Y1 THEN PRINT#1,"R",F
N T(X),Y1:GOTO 340 <004>
330 IF FN T(FN A(X))<Y2 THEN PRINT#1,"R",F
N T(X),Y2:GOTO 340 <169>
335 PRINT#1,"J",FN T(X),FN T(FN A(X)) <036>
340 NEXT X <034>
345 IF (XO-XU)>25 THEN 405 <237>
350 REM +++ PLOTTER EINHEITENZAHLN +++ <243>
355 PRINT#2,3:PRINT#3,0:DX=(XO-XU)/28 <033>
360 FOR I=XU+1 TO XO-1:M=INT(I):PRINT#1,"M
",FN T(-XU),-FN T(YO):PRINT#1,"I" <109>
365 PRINT#1,"R",FN T(M),FN T(-DX):IF M=0 T
HEN 375 <106>
370 PRINT#4,M; <176>
375 NEXT I <205>
380 FOR I=YU+1 TO YO-1:M=INT(I):PRINT#1,"M
",FN T(-XU),-FN T(YO):PRINT#1,"I" <165>
385 PRINT#1,"R",FN T(DX),FN T(M):IF M=0 TH

```

## Listing 8. Funktionenplot.

Erspar die Wertetabelle: Das Programm Funktionenplot

```

EN 395 <011>
390 PRINT#4,M; <196>
395 NEXT I <225>
400 REM +++ PLOTTE FUNKTIONSNAMEN +++ <228>
405 PRINT#2,2 <146>
410 PRINT#1,"M",FN T(-XU),-FN T(YO):PRINT# <209>
1,"I":PRINT#1,"R",X1,Y2
415 PRINT#1,"J",X2,Y2:PRINT#1,"J",X2,Y1:PR <220>
INT#1,"J",X1,Y1:PRINT#1,"J",X1,Y2
420 PRINT#4:PRINT#4:PRINT#2,1:PRINT#3,1:PR <058>
INT#4,"Y" = "F$":PRINT#4:PRINT#2,0
425 REM +++ PLOTTERKANALE SCHLIESSEN + <166>
430 CLOSE 1:CLOSE 2:CLOSE 3:CLOSE 4 <081>
435 END <183>

```

© 64'er

## Listing 8. Funktionenplot (Schluß)

```

1 REM ***** <132>
2 REM * * <051>
3 REM * 3D-GRAFIK MITTELS HIRES-3 * <161>
4 REM * * <053>
5 REM * HEIMO PONNATH HAMBURG 1985 * <130>
6 REM * * <055>
7 REM * HIRES-3 MUSS IM SPEICHER SEIN * <033>
8 REM ***** <139>
9 REM <071>
10 REM ++ INITIALISIEREN DER GRAFIK ++ <036>
15 REM <077>
20 POKE 52,112:POKE 56,112:SYS 37498:GOTO <226>
60 <087>
25 REM <180>
30 REM ++ UP CURSOR SETZEN ++++++ <097>
35 REM
40 POKE 211,SP:POKE 214,ZL:SYS 58640:RETUR <048>
N
45 REM <107>
50 REM +++ EINGABE DER 3D-FUNKTION +++ <051>
55 REM <117>
60 PRINT CHR$(147):SYS 34647:POKE 646,14:Z <196>
L=5:SP=5:GOSUB 40
65 PRINT"FUNKTION IM PROGRAMM:" <089>
70 K=1:GOSUB 180:K=0 <104>
75 PRINT:PRINT TAB(3)"Y"="F$":PRINT <010>
80 PRINT TAB(5)CHR$(18)"A"CHR$(146)"LTE OD <207>
ER "CHR$(18)"N"CHR$(146)"EUE FUNKTION?
85 GET A$:IF A$<>"A"AND A$<>"N"THEN 85 <249>
90 IF A$="A"THEN 140 <043>
95 PRINT CHR$(147):SP=3:ZL=12:GOSUB 40:INP <255>
UT"Y"="F$":POKE 646,6
100 PRINT CHR$(147)CHR$(17)CHR$(17)"180F$= <081>
"CHR$(34)F$CHR$(34) <034>
105 PRINT"190DEFFNA(X)=""F$ <118>
110 PRINT"RUN140":PRINT CHR$(19);
115 POKE 631,13:POKE 632,13:POKE 633,13:PO <073>
KE 198,3:END <182>
120 REM <021>
125 REM +++ NEUBEGINN DES PROGRAMMES ++ <006>
130 REM +++ FALLS NEUE FUNKTION ++ <197>
135 REM <012>
140 PRINT CHR$(147):POKE 646,14
145 REM <207>
150 REM +++ FUNKTIONEN UND VARIABLE +++ <150>
155 REM <217>
160 X=0:Z=0:SP=0:ZL=0:XU=0:XO=0:YU=0:YO=0: <163>
Z3=0:Z4=0:Z5=0:Z6=0:ZU=0:ZO=0
165 Z1=0:Z2=0:F1=1:F2=6:A=1:XA=0:XE=0:ZA=0 <129>
:ZE=0:DZ=1:XT=0:XH=0:YT=0:YH=0:ZZ=1
170 REM <232>
175 A$="" <204>
180 F$="SIN(X)*SIN(Z)" <051>
185 REM <247>
190 DEF FN A(X)=SIN(X)*SIN(Z) <113>
195 DEF FN Z(Z)=Z/(2*SQR(2)) <029>
200 IF K=1 THEN RETURN <209>
205 REM <011>
210 REM ++ EINGABE KOORDINATENSYSTEM ++ <144>
215 REM <021>
220 SP=1:ZL=3:GOSUB 40 <111>
225 PRINT CHR$(18)"UNSER SYSTEM{4SPACE}:"C <085>
HR$(146):SP=7:ZL=5:GOSUB 40
230 INPUT"XU,XO=";XU,XO:ZL=6:GOSUB 40:INPU <176>
T"YU,YO=";YU,YO:SP=9:ZL=8:GOSUB 40
235 Z3=2*XO*SQR(2):Z4=2*YO*SQR(2):IF Z3<Z4 <121>
THEN PRINT"ZO MAXIMAL="Z3:GOTO 245
240 PRINT"ZO MAXIMAL="Z4 <101>

```

```

245 SP=9:ZL=10:GOSUB 40 <064>
250 Z5=2*XU*SQR(2):Z6=2*YO*SQR(2):IF Z5>Z6 <094>
THEN PRINT"ZU MINIMAL="Z5:GOTO 260
255 PRINT"ZU MINIMAL="Z6 <049>
260 SP=7:ZL=12:GOSUB 40:INPUT"ZU,ZO=";ZU,Z <051>
O:Z1=FN Z(ZO):Z2=FN Z(ZU)
265 IF Z1>XO OR Z1>YO THEN 235 <227>
270 IF Z2<XU OR Z2<YU THEN 235 <016>
275 REM <083>
280 REM +++ EINGABE ZEICHENPARAMETER ++ <165>
285 REM <093>
290 ZL=14:SP=1:GOSUB 40:PRINT CHR$(18)"UNS <115>
ERE ZEICHNUNG:"CHR$(146)
295 ZL=16:SP=3:GOSUB 40:INPUT"ZEICHEN- UND <047>
HINTERGRUNDFARBE=";F1,F2
300 ZL=18:SP=7:GOSUB 40:INPUT"SCHRIITWEITE <115>
VON Z=";A
305 ZL=20:GOSUB 40:INPUT"X-BEREICH XA,XE=" <056>
;XA,XE
310 ZL=21:GOSUB 40:INPUT"Z-BEREICH ZA,ZE=" <015>
;ZA,ZE
315 REM <123>
320 REM +++ ZEICHNEN +++ <049>
325 REM <133>
330 DZ=A/2:XT=XU:XH=XO:YT=YU:YH=YO <015>
335 HFL,F1,F2:TRS,XU,XO,YU,YO <232>
340 REM <148>
345 REM ----- KOORDINATENKREUZ ----- <141>
350 REM <158>
355 TLN,XU,0,XO,0:TLN,0,YU,0,YO:TLN,Z2,Z2, <011>
Z1,Z1 <168>
360 REM <208>
365 REM ----- Z-SCHLEIFE ----- <178>
370 REM
375 FOR Z=ZA TO ZE STEP DZ:ZZ=FN Z(DZ):XT= <160>
XT-ZZ:XH=XH-ZZ:YT=YT-ZZ:YH=YH-ZZ
380 TRS,XT,XH,YT,YH <215>
385 FUNKT,A,XA,XE <224>
390 NEXT Z <100>
395 TEX,"Y"="F$,0,3 <191>
400 REM <208>
405 REM ----- ZEICHNUNG FERTIG ----- <071>
410 REM <218>
415 GET A$:IF A$=""THEN 415 <068>
420 HOF <165>
425 REM <233>
430 REM +++++ MENUE +++++ <188>
435 REM <243>
440 ZL=23:SP=1:GOSUB 40 <158>
445 PRINT CHR$(18)"E"CHR$(146)"NDE {4SPACE} <103>
"CHR$(18)"G"CHR$(146)"RAFIK"
450 ZL=23:SP=18:GOSUB 40 <166>
455 PRINT CHR$(18)"N"CHR$(146)"EU {5SPACE}" <229>
CHR$(18)"D"CHR$(146)"RUCKEN"
460 GET A$:IF A$<>"E"AND A$<>"G"AND A$<>"N <043>
"AND A$<>"D"THEN 460 <017>
465 REM <083>
470 REM ----- OPTION GRAFIK ----- <027>
475 REM
480 IF A$="G"THEN:HAN:POKE 198,0:WAIT 198, <109>
1:HOF:GOTO 460
485 REM <037>
490 REM ----- OPTION DRUCKEN ----- <033>
495 REM <047>
500 IF A$="D"THEN:HAN:OPEN 1,4,10:PRINT#1: <069>
CLOSE 1:SYS 34865:HOF
505 IF A$="D"THEN OPEN 1,4:PRINT#1:PRINT#1 <009>
,"XU"=XU,"XO"=XO,"YU"=YU,"YO"=YO
510 IF A$="D"THEN PRINT#1,"ZU"=ZU,"ZO"=ZO, <239>
"A"=A
515 IF A$="D"THEN PRINT#1,"XA"=XA,"XE"=XE, <059>
"ZA"=ZA,"ZE"=ZE:PRINT#1:CLOSE 1:GOTO 4
60 <074>
520 REM <190>
525 REM --- OPTION NEUE ZEICHNUNG --- <084>
530 REM <053>
535 IF A$="N"THEN RUN <094>
540 REM <127>
545 REM --- OPTION PROGRAMME --- <104>
550 REM <049>
555 END <202>
560 REM ***** DAS WARS ! *****

```

© 64'er

## Listing 9. 3D-Programm.

Schnell geworden: Das 3D-Programm aus der Grafik-Serie mit HiRes-3.